
import nsrunit; unit conversion on;
math SAhalfLife {
// INDEPENDENT VARIABLE
realDomain t hour; t.min=0; t.max=16.0; t.delta=0.05;

// PARAMETERS
real Alo = 1 mg/L, Amid =50 mg/L, Ahi =300 mg/L;    // Initial Concentrations
real thalflo =5 hour, thalfmid = 5 hour, thalfhi = 5 hour;  // Half-life

// DEPENDENT VARIABLES
real lo(t)   mg/L,   mid(t) mg/L,    hi(t) mg/L;    // Concentrations

// INITIAL CONDITIONS
when(t=t.min) {lo=Alo; mid=Amid; hi=Ahi; }

// ORDINARY DIFFERENTIAL EQUATIONS
lo:t=(-ln(2)/thalflo)*lo;
mid:t=(-ln(2)/thalfmid)*mid;
hi:t=(-ln(2)/thalfhi)*hi;
}
__________________________________________________________________________

import nsrunit; unit conversion on;
math SAenzyme {

// INDEPENDENT VARIABLE
realDomain t hour; t.min=0; t.max=16.0; t.delta=0.05;

/* Enzyme model for all sets of data
   using shared parameters

           kf1                  kf2
           -->                  -->
  SA+Enzyme     SAEnzymeComplex     Enzyme+Product
           <--                  <--
           kb1                  kb2 = 0

*/
// PARAMETERS
real Alo = 1 mg/L, Amid =50 mg/L, Ahi =300 mg/L,  // Initial Concentrations
// SHARED PARAMETERS
     kf1  = 1 L*mg^(-1)*hour^(-1),                // Rate constants
     kb1  = 1 hour^(-1),
     kf2  = 1 hour^(-1),
     Etot = 6 mg/L;                              // Total enzyme

// VARIABLES
real LO(t)  mg/L,  LOE(t)  mg/L,  Elo(t)  mg/L;
real MID(t) mg/L,  MIDE(t) mg/L,  Emid(t) mg/L;
real HI(t)  mg/L,  HIE(t)  mg/L,  Ehi(t)  mg/L;

// INITIAL CONDITIONS
when(t=t.min) {LO = Alo;  MID = Amid; HI = Ahi;
               LOE   = 0;    MIDE   = 0;    HIE   = 0;}
// ORDINARY DIFFERENTIAL EQUATIONS
LO:t  =   -kf1*LO*Elo+kb1*LOE;
LOE:t    = -(-kf1*LO*Elo+kb1*LOE+kf2*LOE);
Elo      = Etot-LOE;

MID:t =   -kf1*MID*Emid+kb1*MIDE;
MIDE:t   = -(-kf1*MID*Emid+kb1*MIDE+kf2*MIDE);
Emid     = Etot-MIDE;



HI:t  =   -kf1*HI*Ehi+kb1*HIE;
HIE:t    = -(-kf1*HI*Ehi+kb1*HIE+kf2*HIE);
Ehi      = Etot-HIE;

// BRIGGS-HALDANE CALCULATION
real Km  mg/L,
     Vmax mg/L*hour^(-1);
Km=(kf2+kb1)/kf1;
Vmax=kf2*Etot;
}
____________________________________________________________________________

import nsrunit; unit conversion on;
math SABriggsHaldane {

// INDEPENDENT VARIABLE
realDomain t hour; t.min=0; t.max=16.0; t.delta=0.05;
/*-----------Briggs-Haldane Enzyme Kinetics-------------

   dC/dt = -Vmax*C/(Km+C)

*/
// PARAMETERS

real Alo = 1 mg/L, Amid =50 mg/L, Ahi =300 mg/L,
     Km = 10 mg/L,
     Vmax = 8 mg/(L*hour);

// VARIABLES
real LO(t) mg/L, MID(t) mg/L, HI(t) mg/L;

// INITIAL CONDITIONS
when(t=t.min) {LO=Alo; MID=Amid; HI=Ahi;}

// ORDINARY DIFFERENTIAL EQUATIONS
LO:t  = -Vmax*LO/(Km+LO);
MID:t = -Vmax*MID/(Km+MID);
HI:t  = -Vmax*HI/(Km+HI);
}


